NAG Fortran Library Routine Document

F02EBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F02EBF computes all the eigenvalues, and optionally all the eigenvectors, of a real general matrix.

2 Specification

```
      SUBROUTINE F02EBF (JOB, N, A, LDA, WR, WI, VR, LDVR, VI, LDVI, WORK,
     1                   LWORK, IFAIL)
      INTEGER            N, LDA, LDVR, LDVI, LWORK, IFAIL
      double precision   A(LDA,*), WR(*), WI(*), VR(LDVR,*), VI(LDVI,*),
     1                   WORK(LWORK)
      CHARACTER*1        JOB
```

3 Description

F02EBF computes all the eigenvalues, and optionally all the right eigenvectors, of a real general matrix A:

$$Ax_i = \lambda_i x_i, \quad i = 1, 2, \ldots, n.$$

Note that even though A is real, λ_i and x_i may be complex. If x_i is an eigenvector corresponding to a complex eigenvalue λ_i, then the complex conjugate vector $\bar{x}_i$ is the eigenvector corresponding to the complex conjugate eigenvalue $\bar{\lambda}_i$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOB – CHARACTER*1 *Input*

On entry: indicates whether eigenvectors are to be computed.

JOB = 'N'

Only eigenvalues are computed.

JOB = 'V'

Eigenvalues and eigenvectors are computed.

Constraint: JOB = 'N' or 'V'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n general matrix A.

On exit: if JOB = 'V', A contains the Schur form of the balanced input matrix A' (see Section 8).

If JOB = 'N', the contents of A are overwritten.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F02EBF is called.

Constraint: LDA $\geq \max(1, \text{N})$.

5: WR(∗) – ***double precision*** array *Output*
6: WI(∗) – ***double precision*** array *Output*

Note: the dimension of the array WR and WI must be at least $\max(1, \text{N})$.

On exit: WR and WI hold the real and imaginary parts, respectively, of the computed eigenvalues. Complex conjugate pairs of eigenvalues are stored in consecutive elements of WR and WI, with the eigenvalue having positive imaginary part first.

7: VR(LDVR,∗) – ***double precision*** array *Output*

Note: the second dimension of the array VR must be at least $\max(1, \text{N})$ if JOB = 'V' and at least 1 otherwise.

On exit: If JOB = 'V', VR contains the real parts of the eigenvectors, with the ith column holding the real part of the eigenvector associated with the eigenvalue λ_i (stored in WR(i) and WI(i)).

If JOB = 'N', VR is not referenced.

8: LDVR – INTEGER *Input*

On entry: the first dimension of the array VR as declared in the (sub)program from which F02EBF is called.

Constraints:

if JOB = 'N', LDVR ≥ 1;
if JOB = 'V', LDVR $\geq \max(1, \text{N})$.

9: VI(LDVI,∗) – ***double precision*** array *Output*

Note: the second dimension of the array VI must be at least $\max(1, \text{N})$ if JOB = 'V' and at least 1 otherwise.

On exit: If JOB = 'V', VI contains the imaginary parts of the eigenvectors, with the ith column holding the imaginary part of the eigenvector associated with the eigenvalue λ_i (stored in WR(i) and WI(i)).

If JOB = 'N', VI is not referenced.

10: LDVI – INTEGER *Input*

On entry: the first dimension of the array VI as declared in the (sub)program from which F02EBF is called.

Constraints:

if JOB = 'N', LDVI ≥ 1;
if JOB = 'V', LDVI $\geq \max(1, \text{N})$.

11: WORK(LWORK) – ***double precision*** array *Workspace*
12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F02EBF is called. On some high-performance computers, increasing the dimension of WORK will enable the routine to run faster; a value of $64 \times \mathrm{N}$ should allow near-optimal performance on almost all machines.

Constraint: LWORK $\geq \max(1, 4 \times \mathrm{N})$.

13: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, JOB $\neq$ 'N' or 'V',
or N < 0,
or LDA $< \max(1, \mathrm{N})$,
or LDVR < 1, or LDVR $<$ N and JOB = 'V',
or LDVI < 1, or LDVI $<$ N and JOB = 'V',
or LWORK $< \max(1, 4 \times \mathrm{N})$.

IFAIL $= 2$

The *QR* algorithm failed to compute all the eigenvalues.

7 Accuracy

If λ_i is an exact eigenvalue, and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$\left|\tilde{\lambda}_i - \lambda_i\right| \leq \frac{c(n)\epsilon\left\|A'\right\|_2}{s_i},$$

where $c(n)$ is a modestly increasing function of n, ϵ is the ***machine precision***, and s_i is the reciprocal condition number of λ_i; A' is the balanced form of the original matrix A (see Section 8), and $\left\|A'\right\| \leq \|A\|$.

If x_i is the corresponding exact eigenvector, and $\tilde{x}_i$ is the corresponding computed eigenvector, then the angle $\theta(x_i, x_i)$ between them is bounded as follows:

$$\theta(x_i, x_i) \leq \frac{c(n)\epsilon\left\|A'\right\|_2}{sep_i}$$

where sep_i is the reciprocal condition number of x_i.

The condition numbers s_i and sep_i may be computed by calling F08QLF (DTRSNA), using the Schur form of the balanced matrix A' which is returned in the array A when JOB = 'V'.

8 Further Comments

F02EBF calls routines from LAPACK in Chapter F08. It first balances the matrix, using a diagonal similarity transformation to reduce its norm; and then reduces the balanced matrix A' to upper Hessenberg form H, using an orthogonal similarity transformation $A' = QHQ^{\mathrm{T}}$. If only eigenvalues are required, the routine uses the Hessenberg QR algorithm to compute the eigenvalues. If the eigenvectors are required, the routine first forms the orthogonal matrix Q that was used in the reduction to Hessenberg form; it then uses the Hessenberg QR algorithm to compute the Schur factorization of A' as $A' = ZTZ^{\mathrm{T}}$. It computes the right eigenvectors of T by backward substitution, pre-multiplies them by Z to form the eigenvectors of A', and finally transforms the eigenvectors to those of the original matrix A.

Each eigenvector x (real or complex) is normalized so that $\|x\|_2 = 1$, and the element of largest absolute value is real and positive.

The time taken by the routine is approximately proportional to n^3.

9 Example

To compute all the eigenvalues and eigenvectors of the matrix A, where

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix}.$$

9.1 Program Text

```
*     F02EBF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA, LDV, LDVI, LDVR, LWORK
      PARAMETER        (NMAX=8,LDA=NMAX,LDV=NMAX,LDVI=NMAX,LDVR=NMAX,
     +                 LWORK=64*NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, N
*     .. Local Arrays ..
      COMPLEX *16      V(LDV,NMAX)
      DOUBLE PRECISION A(LDA,NMAX), VI(LDVI,NMAX), VR(LDVR,NMAX),
     +                 WI(NMAX), WORK(LWORK), WR(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         F02EBF, X04DBF
*     .. Intrinsic Functions ..
      INTRINSIC        CMPLX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F02EBF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*        Compute eigenvalues and eigenvectors of A
*
         IFAIL = 0
*
         CALL F02EBF('Vectors',N,A,LDA,WR,WI,VR,LDVR,VI,LDVI,WORK,LWORK,
     +               IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Eigenvalues'
         WRITE (NOUT,99999) (' (',WR(I),',',WI(I),')',I=1,N)
         WRITE (NOUT,*)
         DO 40 J = 1, N
            DO 20 I = 1, N
               V(I,J) = CMPLX(VR(I,J),VI(I,J),KIND=KIND(A))
```

```
   20       CONTINUE
   40    CONTINUE
*
         CALL X04DBF('General',' ',N,N,V,LDV,'Bracketed','F7.4',
     +               'Eigenvectors','Integer',RLABS,'Integer',CLABS,80,
     +               0,IFAIL)
*
      END IF
      STOP
*
99999 FORMAT ((3X,4(A,F7.4,A,F7.4,A,:)))
      END
```

9.2 Program Data

```
F02EBF Example Program Data
  4                            :Value of N
  0.35   0.45  -0.14  -0.17
  0.09   0.07  -0.54   0.35
 -0.44  -0.33  -0.03   0.17
  0.25  -0.32  -0.13   0.11   :End of matrix A
```

9.3 Program Results

```
 F02EBF Example Program Results

 Eigenvalues
    ( 0.7995, 0.0000) (-0.0994, 0.4008) (-0.0994,-0.4008) (-0.1007, 0.0000)

 Eigenvectors
                    1                  2                  3                  4
 1  ( 0.6551, 0.0000) (-0.1933, 0.2546) (-0.1933,-0.2546) ( 0.1253, 0.0000)
 2  ( 0.5236, 0.0000) ( 0.2519,-0.5224) ( 0.2519, 0.5224) ( 0.3320, 0.0000)
 3  (-0.5362, 0.0000) ( 0.0972,-0.3084) ( 0.0972, 0.3084) ( 0.5938, 0.0000)
 4  ( 0.0956, 0.0000) ( 0.6760, 0.0000) ( 0.6760, 0.0000) ( 0.7221, 0.0000)
```